75 Newport Road, Suite 211
New London, NH 03257
Voice: (603) 526-9088
Fax: (603) 526-9004
info@canarysystems.com
www.canarysystems.com

Triggering Measurements using a Kinemetrics Altus Digital Recorder

MultiLogger Application Note #9

Overview

The Kinemetrics Altus Digital Recorder is designed to profile seismic activity by monitoring the outputs of highly sensitive triaxial accelerometers. The Altus Digital Recorder includes a digital output port that can be used to trigger monitoring of secondary instruments, this Application Note will outline the hardware and software issues related to utilizing this output to signal measurements for instrument connected to a Campbell based system which has been programmed using MultiLogger.

Refer to the Altus Digital Recorder User Manual for more information on the Kinemetrics recorder.
Refer to the QuickTalk & QuickLook Communications Software User Guide for more information on the Kinemetrics software.
Refer to the Campbell Scientific CR10X Operators Manual for more information on the Control Module.

Hardware Interface

The Altus includes an ALARMOUT pin (open-collector type output) on the Auxiliary I/O connector, as described in chapter 5 of the Altus Digital Recorder User Manual on page 122. This output is activated synchronously with the trigger of sampling, as defined by the user.

This output is connected (using the interface circuit shown below) to control port 7 on the Campbell Control Module to signal for measurements. Contact Canary Systems or your systems vendor to obtain the required interface cable and circuit shown below. See page 152 of the Altus User Manual for more information on the Alarm Output signal.



This circuit will incur approximately 5mA of quiescent current, this should be considered as it relates to the operating time of the system while operating solely on battery power.

This circuit also provides optical isolation between the Altus Recorder and CR10X Control Module.

Altus Digital Recorder Configuration

By default the Altus Digital Recorder will set the alarm state at the first triggering and it will remain there until manually reset. **This means that subsequent activation of the measurements will NOT BE PERFORMED!** You must configure the Altus to automatically reset the alarm output so that subsequent triggers will be activated. This is accomplished by setting the **Alarm Duration** parameter of the Altus Recorder. It is also a good idea to manually reset the alarm prior to setting the Alarm Duration parameter. Both of these commands are issued using the Kinemetrics QuickTalk software supplied with the Recorder.

Connect the PC to the Recorder, start the QuickTalk software, the connection can be verified by pressing the **SetClock** button. Select the menu option **Window | Terminal Window**, you should see a terminal emulation window display. Press <ENTER> a few times, you should see the * prompt returned from the Altus Recorder. If not consult the QuickTalk User Guide for troubleshooting information.

Type the command **CLEAR ALARM** <ENTER>. You should see the * prompt returned.

Type the command **ALARM DURATION nn** <ENTER>, where nn is the duration in seconds that the Alarm Output line is held active before reset. A value of 5 seconds is adequate. You should see the * prompt returned.

MultiLogger Configuration

The configuration of MultiLogger is quite straightforward, it involves basically three steps.

The first step is to activate the use of a special subroutine that will set a software flag in the event of the hardware signal being received from the Altus recorder.

This configuration is performed using the **Program | Tables** menu item in the Logger configuration form of MultiLogger. Be sure to check the respective checkbox, the file is selected using the file open button located to the right of each edit. See the screenshot at the right.



Note: The **Altus_Table3.ins** file will reside in your MultiLogger program directory, this option is installed with version 2.0.12 or higher of MultiLogger. Contact Canary Systems or your software vendor to obtain assistance installing this file with an older version.

The second step is to assign a location to store the number of instrument scans to be performed in the event of a trigger from Altus Recorder. This configuration is performed using the **Program | Input Locations** menu item in the Logger configuration form of MultiLogger. It is recommended to use one of the available User Locations, as shown in the screenshot at the right.



The third step is to select the **Altus Alarm** Data Output option. This option configures three modes of data storage.

- **Data are stored at the default interval if any channel exceeds alarm settings.**
- **Data are stored at the default interval for 24 hours if the Altus triggers.**
- **Data are stored 4 times a day, at Midnight, 6AM, Noon and 6PM, in the quiescent state.**

Note: The **Altus Alarm** option is installed with version 2.0.12 or higher of MultiLogger. Contact Canary Systems or your software vendor to obtain assistance installing this file with an older version.

Final Note: Set the **Single Interval** to 60 seconds for optimum performance

Additional operation details regarding these files are contained in the following sections.

Altus Alarm Instruction File

The instruction file can be loaded into the editor by clicking the gear button located to the left of the selected Data Output option. The file contents are listed below, note in particular the ability to configure the default measurement interval and repetitions of measurements in the event of a trigger from the Altus Digital Recorder.

```
;Make sure ports are configured to accept alarm
P20     Set Port(s)     ;
1:[9899                 ]       C8,C7,C6,C5 Options     ;
2:[9999                 ]       C4,C3,C2,C1 Options     ;

;Check if Altus alarm
P91     If Flag/Port    ;
1:[12                   ]       Flag/Port Options (Do if Flag 2 is High)      ;
2:[30                   ]       Command Code Option (Then Do) ;

P86     Do      ;
1:[StoreSub             ]       Command Code Option     ;

;Make sure alarm flag is set high to trigger dialer
P86     Do      ;
1:[18                   ]       Command Code Option (Set Flag 8 High) ;

;Keep count of alarm flags
P32     Z=Z+1   ;
1:[AltusCount           ]       Z Loc   ;

P89     If (X<=>F)      ;
1:[AltusCount           ]       X Loc   ;
2:[3                    ]       Comparison Code Option (>=)   ;
3:[1440                 ]       F       ; set number of scans after trigger
4:[30                   ]       Command Code Option (Then Do) ;

;Reset our flag
P86     Do      ;
1:[22                   ]       Command Code Option (Set Flag 2 Low)  ;

;Reset our counter
P30     Z=F     ;
1:[0                    ]       F       ;
2:[0                    ]       Exponent of 10 ;
3:[AltusCount           ]       Z Loc   ;

P95     End     ;

P94     Else    ;

;Reset our Altus counter
```

```
P30    Z=F    ;
1:[0                ]    F       ;
2:[0                ]    Exponent of 10 ;
3:[AltusCount       ]    Z Loc   ;

;Check if measurement alarm
P91    If Flag/Port   ;
1:[18               ]    Flag/Port Options (Do if Flag 8 is High)     ;
2:[30               ]    Command Code Option (Then Do) ;

P86    Do     ;
1:[StoreSub         ]    Command Code Option    ;

P94    Else   ;

;Output Data at 6am
P92    If time is     ;
1:[360              ]    Minutes (Seconds --) into a   ;
2:[1440             ]    Interval (same units as above)       ;
3:[StoreSub         ]    Command Code Option    ;

;Output Data at Noon
P92    If time is     ;
1:[720              ]    Minutes (Seconds --) into a   ;
2:[1440             ]    Interval (same units as above)       ;
3:[StoreSub         ]    Command Code Option    ;

;Output Data at 6pm
P92    If time is     ;
1:[1080             ]    Minutes (Seconds --) into a   ;
2:[1440             ]    Interval (same units as above)       ;
3:[StoreSub         ]    Command Code Option    ;

;Output Data at Midnight
P92    If time is     ;
1:[0                ]    Minutes (Seconds --) into a   ;
2:[1440             ]    Interval (same units as above)       ;
3:[StoreSub         ]    Command Code Option    ;

P95    End    ;

P95    End    ;
```

Altus Table 3 Instruction File (should not be modified by the user)

```
P85    Beginning of Subroutine       ;
1:[97               ]    Subroutines (Subroutine 97)   ;

P86    Do     ;
1:[12               ]    Command Code Option (Set Flag 2 High) ;

P95    End    ;
```